Prolog végrehajtás LP-117

Visszalépéses keresés — számintervallum felsorolása

dec(J) felsorolta a 0 és 9 közötti egész számokat

Általánosítás: soroljuk fel az N és M közötti egészeket (N és M maguk is egészek)

% between(M, N, I): M =< I =< N, I egész.
between(M, N, M) :-
        M =< N.
between(M, N, I) :-
        M < N,
        M1 is M+1,
        between(M1, N, I).

% dec(X): X egy decimális számjegy
dec(X) :- between(0, 9, X).

| ?- between(1, 2, _X), between(3, 4, _Y), Z is 10*_X+_Y.
Z = 13 ? ;
Z = 14 ? ;
Z = 23 ? ;
Z = 24 ? ;
no

Prolog végrehajtás LP-118

A SICStus eljárás-doboz alapú nyomkövetése — legfontosabb parancsok

Alapvető nyomkövetési parancsok

h <RET> (help) — parancsok listázása
c <RET> (creep) vagy <RET> — továbblépés minden kapunál megálló nyomkövetéssel
l <RET> (leap) — csak töréspontnál áll meg, de a dobozokat építi
z <RET> (zip) — csak töréspontnál áll meg, dobozokat nem épít
+ <RET> ill. - <RET> — töréspont rakása/eltávolítása a kurrens predikátumra
s <RET> (skip) — eljárástörzs átlépése (Call/Redo ⇒ Exit/Fail)
o <RET> (out) — kilépés az eljárástörzsből

A Prolog végrehajtást megváltoztató parancsok

u <RET> (unify) — a kurrens hívást végrehajtás helyett egyesíti egy beolvasott kifejezéssel.
r <RET> (retry) — újrakezdi a kurrens hívás végrehajtását (ugrás a Call kapura)

Információ-megjelenítő és egyéb parancsok

w <RET> (write) — a hívás kiírása mélység-korlátozás nélkül
b <RET> (break) — új, beágyazott Prolog interakciós szint létrehozása
n <RET> (notrace) — nyomkövető kikapcsolása
a <RET> (abort) — a kurrens futás abbahagyása

Deklaratív programozás. BME VIK, 2003. tavaszi félév (Logikai Programozás)

TOVÁBBI VEZÉRLÉSI SZERKEZETEK

További vezérlési szerkezetek LP-120

Negáció

Korábbi feladat:

Az 1, 3, 4, 6 számokból a négy alapművelet felhasználásával állítsuk elő a 24 számértéket!

Érdekes kérdés: melyik az első természetes szám, amely nem áll elő pl. az 1, 3, 4, 6 számokból a négy alapművelet felhasználásával?

Ehhez negációra van szükségünk: a A \+ Hívás akkor és csak akkor sikerül, ha Hívás meghiúsul.

| ?- between(1, 1000, E), \+ negylevelu_erteke(1,3,4,6, E, _).
E = 34 ? ;
E = 38 ? ;
E = 39 ? ;
E = 44 ? ...

További vezérlési szerkezetek LP-121

A meghiúsulásos negáció (NF — Negation by Failure)

A \+ Hívás beépített meta-eljárás (vö. ⊬ — nem bizonyítható) végrehajtja a Hívás hívást, ha Hívás sikeresen lefutott, akkor meghiúsul, egyébként (azaz ha Hívás meghiúsult) sikerül.

\+ Hívás futása során Hívás legfeljebb egy megoldása áll elő

\+ Hívás sohasem helyettesít be változót

Gondok a meghiúsulásos negációval:

„zárt világ feltételezése” (CWA) — ami nem bizonyítható, az nem igaz.

| ?- \+ szuloje(’Imre’, X). ----> no
| ?- \+ szuloje(’Géza’, X). ----> true ?

\ + H deklaratív szemantikája: ¬∃X(H), ahol X a H-ban a hívás pillanatában behelyettesítetlen változókat jelöli.

| ?- \+ X = 1, X = 2. ----> no
| ?- X = 2, \+ X = 1. ----> X = 2 ?

További vezérlési szerkezetek LP-122

Példa: együttható meghatározása lineáris kifejezésben

Formula: számokból és az ‘x’ névkonstansból ‘+’ és ‘*’ operátorokkal épül fel.

% :- type kif == {x} \/ number \/ {kif+kif} \/ {kif*kif}.

Lineáris formula: a ‘*’ operátor legalább egyik oldalán szám áll.

% egyhat(Kif, E): A Kif lineáris formulában az x együtthatója E.
egyhat(x, 1). egyhat(K1*K2, E) :-
egyhat(Kif, E) :- number(K1),
        number(Kif), E = 0. egyhat(K2, E0),
egyhat(K1+K2, E) :- E is K1*E0.
        egyhat(K1, E1), egyhat(K1*K2, E) :-
        egyhat(K2, E2), number(K2),
        E is E1+E2. egyhat(K1, E0),
                                                    E is K2*E0.

| ?- egyhat(((x+1)*3)+x+2*(x+x+3), E). | ?- egyhat(2*3+x, E).
E = 8 ? ; E = 1 ? ;
no E = 1 ? ; no

További vezérlési szerkezetek LP-123

Együttható meghatározása: többszörös megoldások kiküszöbölése

negáció alkalmazásával:

(...)
egyhat(K1*K2, E) :-
        number(K1), egyhat(K2, E0), E is K1*E0.
egyhat(K1*K2, E) :-
        \+ number(K1),
        number(K2), egyhat(K1, E0), E is K2*E0.

hatékonyabban, feltételes kifejezéssel:

(...)
egyhat(K1*K2, E) :-
        ( number(K1) -> egyhat(K2, E0), E is K1*E0
        ; number(K2), egyhat(K1, E0), E is K2*E0
        ).

További vezérlési szerkezetek LP-124

Feltételes kifejezések

Szintaxis (felt, akkor, egyébként tetszőleges célsorozatok):

(...) :-
        (...),
        ( felt -> akkor
        ; egyébként
        ),
        (...).

Deklaratív szemantika: a fenti alak jelentése megegyezik az alábbival, ha a felt egy egyszerű feltétel (nem oldható meg többféleképpen):

(...) :-
        (...),
        ( felt, akkor
        ; \+ felt, egyébként
        ),
        (...).

További vezérlési szerkezetek LP-125

Feltételes kifejezések (folyt.)

Procedurális szemantika A (felt->akkor;egyébként),folytatás célsorozat végrehajtása:

Végrehajtjuk a felt hívást.

Ha felt sikeres, akkor az akkor,folytatás célsorozatra redukáljuk a fenti célsorozatot, a

felt első megoldása által eredményezett behelyettesítésekkel. A felt cél többi megoldását

nem keressük meg.

Ha felt sikertelen, akkor az egyébként,folytatás célsorozatra redukáljuk, behelyettesítés nélkül.

Többszörös elágaztatás skatulyázott feltételes kifejezésekkel:

( felt1 -> akkor1 ( felt1 -> akkor1
; felt2 -> akkor2 ; (felt2 -> akkor2
; ... ; ...
) ...))

Az egyébként rész elhagyható, alapértelmezése: fail.

További vezérlési szerkezetek LP-126

Feltételes kifejezés — példák

Faktoriális

% fakt(+N, ?F): N! = F.
fakt(N, F) :-
        ( N = 0 -> F = 1 % N = 0, F = 1
        ; N > 0, N1 is N-1, fakt(N1, F1), F is N*F1
        ).

Jelentése azonos a sima diszjunkciós alakkal (lásd komment), de annál hatékonyabb, mert nem hagy maga után választási pontot.

Szám előjele

% Sign = sign(Num)
sign(Num, Sign) :-
        ( Num > 0 -> Sign = 1
        ; Num < 0 -> Sign = -1
        ; Sign = 0
        ).

További vezérlési szerkezetek LP-127

Feltételes kifejezés és negáció

A \+ felt negáció kiváltható a ( felt -> fail ; true ) feltételes kifejezéssel.

Példa: ellenőrizzük, hogy egy adott szám nem levele egy fának

nem_levele(Fa, V) :-
        ( fa_levele(F, V) -> fail
        ; true
        ).

(Ismétlés:) A \= beépített eljárás jelentése: az argumentumok nem egyesíthetők, megvalósítása:

X \= Y :- \+ X = Y.

A nem-levele példa-eljárás megvalósítható rekurzívan is:

nem_levele(leaf(V0), V) :-
        V0 \= V.
nem_levele(node(L, R), V) :-
        nem_levele(L, V),
        nem-levele(R, V).

A diszjunktív (exisztenciális kvantornak megfelelő) fa_levele eljárás negáltja egy konjunktív (univerzális kvantoros) bejárás!

LISTÁK PROLOGBAN

A Prolog lista LP-129

A Prolog lista-fogalma

közönséges adattípus: % :- type list(T) ---> .(T,list(T)) ; [].

T típusú elemekből álló lista az vagy egy ’.’/2 struktúra, vagy a [] névkonstans. A struktúra első argumentuma T típusú, a lista feje (első eleme). A második argumentum list(T) típusú, a lista farka (a többi elemből álló lista); egyszerűsített írásmód („szintaktikus édesítés”); hatékonyabb megvalósítás.

A listák fastruktúra alakja és megvalósítása

Elem_2 Farok_2

Farok_1

Elem_1 .(Elem_1, Farok_1)

Elem_N

. . .

NULL [ ]

A Prolog lista LP-130

Listák jelölése — szintaktikus édesítőszerek

  [Fej|Farok] ≡ .(Fej, Farok)

  [Elem_1,Elem_2,...,Elem_N |Farok] ≡ [Elem_1|[Elem_2,...,Elem_N |Farok]]
  [Elem_1,Elem_2,...,Elem_N ] ≡ [Elem_1,Elem_2,...,Elem_N |[]]

| ?- [1,2] = [X|Y]. ⇒ X = 1, Y = [2] ?
| ?- [1,2] = [X,Y]. ⇒ X = 1, Y = 2 ?
| ?- [1,2,3] = [X|Y]. ⇒ X = 1, Y = [2,3] ?
| ?- [1,2,3] = [X,Y]. ⇒ no
| ?- [1,2,3,4] = [X,Y|Z]. ⇒ X = 1, Y = 2, Z = [3,4] ?
| ?- L = [1|_], L = [_,2|_]. ⇒ L = [1,2|_A] ? % nyílt végű
| ?- L = .(1,[2,3|[]]). ⇒ L = [1,2,3] ?
| ?- L = [1,2|.(3,[])]. ⇒ L = [1,2,3] ?
| ?- [X|[3-Y/X|Y]]= .(A, [A-B,6]). ⇒ A=3, B=[6]/3, X=3, Y=[6] ?

A Prolog lista LP-131

Tömör és minta-kifejezések, lista-minták, nyílt végű listák

(Ismétlés:) Tömör (ground) kifejezés: változót nem tartalmazó kifejezés

Minta: egy általában nem nem tömör kifejezés, mindazon kifejezéseket „képviseli”, amelyek belőle változó-behelyettesítéssel előállnak.

Lista-minta: listát (is) képviselő minta.

Nyílt végű lista: olyan lista-minta, amely bármilyen hosszú listát is képvisel.

Zárt végű lista: olyan lista(-minta), amely egyféle hosszú listát képvisel.

Zárt végű Milyen listákat képvisel

[X] egyelemű
[X,Y] kételemű
[X,X] két egyforma elemből álló
[X,1,Y] 3 elemből áll, 2. eleme 1

Nyílt végű Milyen listákat képvisel

X tetszőleges
[X|Y] nem üres (legalább 1 elemű)
[X,Y|Z] legalább 2 elemű
[a,b|Z] legalább 2 elemű, elemei: a, b, ...

A Prolog lista LP-132

Listák összefűzése: az append/3 eljárás

append(L1, L2, L3): Az L3 lista az L1 és L2 listák elemeinek egymás után fűzésével áll elő

(jelöljük: L3 = L1⊕L2) — két megoldás:

append0([], L2, L) :- L = L2.
append0([X|L1], L2, L) :-
    append0(L1, L2, L3), L = [X|L3].

                                          append([], L, L).
                                          append([X|L1], L2, [X|L3]) :-
                                              append(L1, L2, L3).

    > append0([1,2,3],[4],A)
(2) > append0([2,3],[4],B), A=[1|B]
(2) > append0([3],[4],C), B=[2|C], A=[1|B]
(2) > append0([],[4],D),C=[3|D],B=[2|C],A=[1|B]
(1) > D=[4], C=[3|D], B=[2|C], A=[1|B]
BIP > C=[3,4], B=[2|C], A=[1|B]
BIP > B=[2,3,4], A=[1|B]
BIP > A=[1,2,3,4]
BIP > []
L = [1,2,3,4] ?

                                                       > append([1,2,3],[4],A), write(A)
                                                   (2) > append([2,3],[4],B), write([1|B])
                                                   (2) > append([3],[4],C), write([1,2|C])
                                                   (2) > append([],[4],D), write([1,2,3|D])
                                                   (1) > write([1,2,3,4])
                                                   [1,2,3,4]
                                                   BIP > []
                                                   L = [1,2,3,4] ?

Az append0/append(L1, ...) komplexitása: futási ideje arányos L1 hosszával.

Miért jobb az append/3 mint az append0/3?

append/3 jobbrekurzív, ciklussal ekvivalens (nem fogyaszt vermet)

append([1,...,1000],[0],[2,...]) azonnal, append0(...) 1000 lépésben hiúsul meg

append/3 használható szétszedésre is (lásd később), míg append0/3 nem.

A Prolog lista LP-133

Lista építése elölről — nyílt végű listákkal

Az append eljárás már az első redukciónál felépíti az eredmény fejét

append([], L, L).
append([X|L1], L2, [X|L3]) :- append(L1, L2, L3).

| ?- append([1,2,3], [4], Ered) =⇒ Ered = [1|A], append([2,3], [4], A)

Haladó nyomkövetési lehetőségek ennek demonstrálására

library(debugger_examples) — példák a nyomkövető programozására, új parancsokra

új parancs: ‘N ⟨ név ⟩’ — fókuszált argumentum elnevezése szabványos parancs: ‘^ ⟨ argszám ⟩’ — adott argumentumra fókuszálás új parancs: ‘P [⟨ név ⟩]’ — adott nevű (ill összes) kifejezés kiiratása

| ?- use_module(library(debugger_examples)).
| ?- trace, append([1,2,3],[4,5,6],A).
        1 1 Call: append([1,2,3],[4,5,6],_543) ? ^ 3
        1 1 Call: ^3 _543 ? N Ered
        1 1 Call: ^3 _543 ? P ⇒ Ered = _543
        2 2 Call: append([2,3],[4,5,6],_2700) ? P ⇒ Ered = [1|_2700]
        3 3 Call: append([3],[4,5,6],_3625) ? P ⇒ Ered = [1,2|_3625]
        4 4 Call: append([],[4,5,6],_4550) ? P ⇒ Ered = [1,2,3|_4550]
        4 4 Exit: append([],[4,5,6],[4,5,6]) ? P ⇒ Ered = [1,2,3,4,5,6]
        3 3 Exit: append([3],[4,5,6],[3,4,5,6]) ?
        2 2 Exit: append([2,3],[4,5,6],[2,3,4,5,6]) ?
        1 1 Exit: append([1,2,3],[4,5,6],[1,2,3,4,5,6]) ?
A = [1,2,3,4,5,6] ? ; no

A Prolog lista LP-134

Listák megfordítása

Naív (négyzetes lépésszámú) megoldás

% nrev(L, R): Az R lista az L megfordítása.
nrev([], []).
nrev([X|L], R) :-
    nrev(L, RL),
    append(RL, [X], R).

Lineáris lépésszámú megoldás

% reverse(R, L): Az R lista az L megfordítása.
reverse(R, L) :- revapp(L, [], R).

% revapp(L1, L2, R): L1 megfordítását L2 elé fűzve kapjuk R-t.
revapp([], R, R).
revapp([X|L1], L2, R) :-
    revapp(L1, [X|L2], R).

A lists könyvtár tartalmazza az append/3 és reverse/2 eljárások definícióját. A könyvtár betöltése:

:- use_module(library(lists)).

A Prolog lista LP-135

append és revapp — listák gyűjtési iránya

Prolog megvalósítás

append([], L, L).
append([X|L1], L2, [X|L3]) :-
    append(L1, L2, L3).

                                        revapp([], L, L).
                                        revapp([X|L1], L2, L3) :-
                                            revapp(L1, [X|L2], L3).

C++ megvalósítás

struct link { link *next;
                char elem;
                link(char e): elem(e) {}
              };
typedef link *list;

 list append(list list1, list list2)
 { list list3, *lp = &list3;
   for (list p=list1; p; p=p->next)
   { list newl = new link(p->elem);
     *lp = newl; lp = &newl->next;
   }
   *lp = list2;
   return list3;
 }

                                        list revapp(list list1, list list2)
                                        { list l = list2;
                                          for (list p=list1; p; p=p->next)
                                          { list newl = new link(p->elem);
                                            newl->next = l; l = newl;
                                          }

                                          return l;
                                        }

A Prolog lista LP-136

Listák szétbontása az append/3 segítségével

                                                     ?- append(A, B, [1,2,3,4]).
                                              A=[]
                                       B=[1,2,3,4] A=[1|A1]

                                        A=[],B=[1,2,3,4]

                                                          ?- append(A1, B, [2,3,4]).

                                                              A1=[2|A2]

                                                              ?- append(A2, B, [3,4]).

                                                   B=[3,4]

                                                       B=[4]

                                                     A2=[]

                                                          A3=[]

                                            B=[2,3,4]

                                                A1=[]

                                                                        A3=[4|A4]

                                                                   ?- append(A3, B, [4]).

                                                                        ?- append(A4, B, []).

                                                                   A2=[3|A3]

                                            A=[1], B=[2,3,4]

                                                 A=[1,2],B=[3,4]

                                                      A=[1,2,3],B=[4]
                                                                          A4=[]
                                                                          B=[]

                                                                   A=[1,2,3,4],B=[]

% append(L1, L2, L3):
% Az L3 lista az L1 és L2
% listák elemeinek egymás
% után fűzésével áll elő.
append([], L, L).
append([X|L1], L2, [X|L3]) :-
    append(L1, L2, L3).

| ?- append(A, B, [1,2,3,4]).
A = [], B = [1,2,3,4] ? ;
A = [1], B = [2,3,4] ? ;
A = [1,2], B = [3,4] ? ;
A = [1,2,3], B = [4] ? ;
A = [1,2,3,4], B = [] ? ;
no

A Prolog lista LP-137

Variációk appendre 1. — Három lista összefűzése

Az append/3 keresési tere véges, ha első és harmadik argumentuma közül legalább az egyik zárt végű lista.

append(L1,L2,L3,L123): L1 ⊕ L2 ⊕ L3 = L123

append(L1, L2, L3, L123) :-
        append(L1, L2, L12), append(L12, L3, L123).

Nem hatékony, pl.: append([1,...,100],[1,2,3],[1], L) 103 helyett 203 lépés! Szétszedésre nem alkalmas — végtelen választási pontot hoz létre

Szétszedésre is alkalmas, hatékony változat

% L1 ⊕ L2 ⊕ L3 = L123, ahol vagy L1 és L2 vagy L123 adottă(zárt végű).
append(L1, L2, L3, L123) :-
        append(L1, L23, L123), append(L2, L3, L23).

Az első append/3 hívás nyílt végű listát állít elő:

| ?- append([1,2], L23, L). ⇒ L = [1,2|L23] ?

A Prolog lista LP-138

Mintakeresés append/3-mal

Párban előforduló elemek

% párban(Lista, Elem): A Lista számlistának Elem olyan
% eleme, amely egy ugyanilyen értékű elemmel szomszédos.
párban(L, E) :-
    append(_, [E,E|_], L).

| ?- párban([1,8,8,3,4,4], E).
        E = 8 ? ; E = 4 ? ; no

Dadogó részek

% dadogó(L, D): D olyan nem üres részlistája L-nek,
% amelyet egy vele megegyező részlista követ.
dadogó(L, D) :-
        append(_, Farok, L),
        D = [_|_],
        append(D, Vég, Farok),
        append(D, _, Vég).

| ?- dadogó([2,2,1,2,2,1], D).
        D = [2] ? ; D = [2,2,1] ? ; D = [2] ? ; no

A Prolog lista LP-139

Keresés listában

member(E, L): E az L lista eleme
 member(Elem, [Elem|_]).
 member(Elem, [_|Farok]) :-
         member(Elem, Farok).

                                   member(Elem, [Fej|Farok]) :-
                                           ( Elem = Fej
                                           ; member(Elem, Farok)
                                           ).

A member/2 felhasználási lehetőségei

Eldöntendő kérdés

| ?- member(2, [1,2,3]). ⇒ yes

Megválaszolandó kérdések

| ?- member(X, [1,2,3]). ⇒ X = 1 ? ; X = 2 ? ; X = 3 ? ; no
| ?- member(X, [1,2,1]). ⇒ X = 1 ? ; X = 2 ? ; X = 1 ? ; no

Vegyes használat, listák metszete

| ?- member(X, [1,2,3]),
     member(X, [5,4,3,2,3]). ⇒ X = 2 ? ; X = 3 ? ; X = 3 ? ; no

Lista elemévé tesz, végtelen választás!

| ?- member(1, L). ⇒ L = [1|_A] ? ; L = [_A,1|_B] ? ;
                                       L = [_A,_B,1|_C] ? ; ...

A member/2 keresési tere véges, ha második argumentuma zárt végű lista.

A Prolog lista LP-140

member/2 általánosítása: select/3

   select(Elem, Lista, Marad): Elemet a Listaból elhagyva marad Marad.

   select(Elem, [Elem|Marad], Marad). % Elhagyjuk a fejet, marad a farok.
   select(Elem, [X|Farok], [X|Marad0]) :-
           select(Elem, Farok, Marad0). % A farokból hagyunk el elemet.

Felhasználási lehetőségek:

| ?- select(1, [2,1,3], L). % Adott elem elhagyása
        L = [2,3] ? ; no
| ?- select(X, [1,2,3], L). % Akármelyik elem elhagyása
        L=[2,3], X=1 ? ; L=[1,3], X=2 ? ; L=[1,2], X=3 ? ; no
| ?- select(3, L, [1,2]). % Adott elem beszúrása!
        L = [3,1,2] ? ; L = [1,3,2] ? ; L = [1,2,3] ? ; no
| ?- select(3, [2|L], [1,2,7,3,2,1,8,9,4]).
                                       % Beszúrható-e 3 az [1,...]-ba
        no % úgy, hogy [2,...]-t kapjunk?
| ?- select(1, [X,2,X,3], L).
        L = [2,1,3], X = 1 ? ; L = [1,2,3], X = 1 ? ; no

A lists könyvtár tartalmazza a member/2 és select/3 eljárások definícióját is.

A select/3 keresési tere véges, ha 2. és 3. argumentuma közül legalább az egyik zárt végű.

A Prolog lista LP-141

Listák permutációja

permutation(Lista, Perm): Lista permutációja a Perm lista.

permutation([], []).
permutation(Lista, [Elso|Perm]) :-
        select(Elso, Lista, Maradek),
        permutation(Maradek, Perm).

Felhasználási példák:

| ?- permutation([1,2], L).
        L = [1,2] ? ; L = [2,1] ? ; no

| ?- permutation([a,b,c], L).
        L = [a,b,c] ? ; L = [a,c,b] ? ; L = [b,a,c] ? ;
        L = [b,c,a] ? ; L = [c,a,b] ? ; L = [c,b,a] ? ;
        no

| ?- permutation(L, [1,2]).
        L = [1,2] ? ;
        végtelen keresési tér

Ha permutation/2-ben az első argumentum ismeretlen, akkor a select hívás keresési tere végtelen!

PROLOG PÉLDÁK: ÚTVONALKERESÉS GRÁFBAN

Prolog példák: útvonalkeresés gráfban LP-143

A jegyzet bevezető példája: útvonalkeresés

A feladat:

Tekintsük (autóbusz)járatok egy halmazát.

Mindegyik járathoz a két végpont és az útvonal hossza van megadva. Írjunk Prolog eljárást, amellyel megállapítható, hogy két pont összeköthető-e pontosan N csatlakozó járattal!

Átfogalmazás: egy súlyozott irányítatlan gráfban két pont közötti utat keresünk. Élek:

% járat(A, B, H): Az A és B városok között van járat, és hossza H km.
járat(’Budapest’, ’Prága’, 515).
járat(’Budapest’, ’Bécs’, 245).
járat(’Bécs’, ’Berlin’, 635).
járat(’Bécs’, ’Párizs’, 1265).

Irányított élek:

% útszakasz(A, B, H): A-ból B-be eljuthatunk egy H úthosszú járattal.
útszakasz(Kezdet, Cél, H) :-
     ( járat(Kezdet, Cél, H)
     ; járat(Cél, Kezdet, H)
     ).

Prolog példák: útvonalkeresés gráfban LP-144

Az útvonalkeresési feladat — folytatás

Adott lépésszámú útvonal (él-sorozat) és hossza:

% útvonal(N, A, B, H): A és B között van (pontosan)
% N szakaszból álló útvonal, amelynek összhossza H.
útvonal(0, Hová, Hová, 0).
útvonal(N, Honnan, Hová, H) :-
    N > 0,
    N1 is N-1,
    útszakasz(Honnan, Közben, H1),
    útvonal(N1, Közben, Hová, H2),
    H is H1+H2.

Futási példa:

| ?- útvonal(2, ’Párizs’, Hová, H).
        H = 1900, Hová = ’Berlin’ ? ;
        H = 2530, Hová = ’Párizs’ ? ;
        H = 1510, Hová = ’Budapest’ ? ;
        no

Prolog példák: útvonalkeresés gráfban LP-145

Körmentes út keresése

Könyvtár betöltése, adott funktorú eljárások importálásával:

:- use_module(library(lists), [member/2]).

Segéd-argumentum: az érintett városok listája, fordított sorrendben

% útvonal_2(N, A, B, H): A és B között van (pontosan)
% N szakaszból álló körmentes útvonal, amelynek összhossza H.
útvonal_2(N, Honnan, Hová, H) :-
    útvonal_2(N, Honnan, Hová, [Honnan], H).

% útvonal_2(N, A, B, Kizártak, H): A és B között van pontosan
% N szakaszból álló körmentes, Kizártak elemein át nem menő H hosszú út.
útvonal_2(0, Hová, Hová, Kizártak, 0).
útvonal_2(N, Honnan, Hová, Kizártak, H) :-
    N > 0, N1 is N-1, útszakasz(Honnan, Közben, H1),
    \+ member(Közben, Kizártak),
    útvonal_2(N1, Közben, Hová, [Közben|Kizártak], H2), H is H1+H2.

Példa-futás:

| ?- útvonal_2(2, ’Párizs’, Hová, H).
        H = 1900, Hová = ’Berlin’ ? ;
        H = 1510, Hová = ’Budapest’ ? ; no

Prolog példák: útvonalkeresés gráfban LP-146

Továbbfejlesztés: körmentes út keresése, útvonal-gyűjtéssel

Az alapötlet: a Kizártak listában gyűlik a (fordított) útvonal.

A rekurzív eljárásban szükséges egy új argumentum, hogy az útvonalat kiadjuk!

:- use_module(library(lists), [member/2, reverse/2]).

% útvonal_3(N, A, B, Út, H): A és B között van (pontosan)
% N szakaszból álló körmentes Út útvonal, amelynek összhossza H.
útvonal_3(N, Honnan, Hová, Út, H) :-
    útvonal_3(N, Honnan, Hová, [Honnan], FÚt, H),
    reverse(FÚt, Út).

% útvonal_3(N, A, B, FÚt0, FÚt, H): A és B között van pontosan
% N szakaszból álló körmentes, FÚt0 elemein át nem menő H hosszú út.
% FÚt = (az A → B útvonal megfordítása) ⊕ FÚt0.
útvonal_3(0, Hová, Hová, FordÚt, FordÚt, 0).
útvonal_3(N, Honnan, Hová, FordÚt0, FordÚt, H) :-
    N > 0, N1 is N-1, útszakasz(Honnan, Közben, H1),
    \+ member(Közben, FordÚt0),
    útvonal_3(N1, Közben, Hová, [Közben|FordÚt0], FordÚt, H2), H is H1+H2.

| ?- útvonal_3(2, ’Párizs’, _, Út, H).
        H = 1900, Út = [’Párizs’,’Bécs’,’Berlin’] ? ;
        H = 1510, Út = [’Párizs’,’Bécs’,’Budapest’] ? ; no

Prolog példák: útvonalkeresés gráfban LP-147

Súlyozott gráf ábrázolása éllistával

A gráf ábrázolása

a gráf élek listája, az él egy három-argumentumú struktúra, argumentumai: a két végpont és a súly.

Típus-definíció

% :- type él ---> él(pont, pont, súly).
% :- type pont == atom.
% :- type súly == integer.
% :- type gráf == list(él).

Példa

hálózat([él(’Budapest’,’Bécs’,245),
         él(’Budapest’,’Prága’,515),
         él(’Bécs’,’Berlin’,635),
         él(’Bécs’,’Párizs’,1265)]).

Prolog példák: útvonalkeresés gráfban LP-148

Ismétlődésmentes útvonal keresése listával ábrázolt gráfban

:- use_module(library(lists), [select/3]).

% útvonal_4(N, G, A, B, L, H): A G gráfban van egy A-ból
% B-be menő N szakaszból álló L út, melynek összhossza H.
útvonal_4(0, _Gráf, Hová, Hová, [Hová], 0).
útvonal_4(N, Gráf, Honnan, Hová, [Honnan|Út], H) :-
        N >0, N1 is N-1,
        select(Él, Gráf, Gráf1),
        él_végpontok_hossz(Él, Honnan, Közben, H1),
        útvonal_4(N1, Gráf1, Közben, Hová, Út, H2),
        H is H1+H2.

% él_végpontok_hossz(Él, A, B, H): Az Él irányítatlan él
% végpontjai A és B, hossza H.
él_végpontok_hossz(él(A,B,H), A, B, H).
él_végpontok_hossz(él(A,B,H), B, A, H).

| ?- hálózat(_Gráf), útvonal_4(2, _Gráf, ’Budapest’, _, Út, H).
        H = 880, Út = [’Budapest’,’Bécs’,’Berlin’] ? ;
        H = 1510, Út = [’Budapest’,’Bécs’,’Párizs’] ? ;
        no

A PROLOG SZINTAXIS

A Prolog szintaxis LP-150

A Prolog szintaxis összefoglalása

A Prolog szintaxis alapelvei Minden programelem kifejezés!

A szükséges összekötő jelek (’,’, ;, :- -->): szabványos operátorok. A beolvasott kifejezést funktora alapján osztályozzuk:

kérdés: ?- Cél.

Célt lefuttatja, és a változó-behelyettesítéseket kiírja (ez az alapértelmezés az ún.

top-level interaktív felületen).

parancs: :- Cél.

A Célt csendben lefuttatja. Pl. deklaráció (operátor, . . . ) elhelyezésére. szabály: Fej :- Törzs. A szabályt felveszi a programba.

nyelvtani szabály: Fej --> Törzs. Prolog szabállyá alakítja és felveszi (lásd a DCG nyelvtan). tényállítás: Minden egyéb kifejezés. Üres törzsű szabályként felveszi a programba.

A Prolog szintaxis LP-151

A Prolog nyelv-változatok

A SICStus rendszer két üzemmódja

iso Az ISO Prolog szabványnak megfelelő.
sicstus Korábbi változatokkal kompatibilis.

Állítása: set_prolog_flag(language, Mód). Különbségek:

szintaxis-részletek, pl. a 0x1ff szám-alak csak ISO módban, beépített eljárások viselkedésének kisebb eltérései.

az eddig ismertetett eljárások hatása lényegében nem változik.

A Prolog szintaxis LP-152

Szintaktikus édesítőszerek — összefoglalás, gyakorlati tanácsok

Operátoros kifejezések alapstruktúra alakra hozása Zárójelezzük be a kifejezést, az operátorok prioritása és fajtája alapján, például -a+b*2 ⇒

((-a)+(b*2)).

Hozzuk az operátoros kifejezéseket alapstruktúra alakra:

(A Inf B) ⇒ Inf(A,B), (Pref A) ⇒ Pref(A), (A Postf) ⇒ Postf(A)

Példa: ((-a)+(b*2)) ⇒ (-(a)+ *(b,2)) ⇒ +(-(a),*(b,2)). Trükkös esetek:

A vesszőt névként idézni kell: pl. (pp,(qq;rr)) ⇒ ’,’(pp,;(qq,rr)).

- Szám ⇒ negatív számkonstans, de - Egyéb ⇒ prefix alak.

Példa. -1+2 ⇒ +(-1,2), de -a+b ⇒ +(-(a),b).

Név(...) ⇒ struktúrakifejezés;
Név (...) ⇒ prefix operátoros kifejezés. Példák:
-(1,2) ⇒ -(1,2) (változatlan), de
- (1,2) ⇒ -(’,’(1,2)) .

A Prolog szintaxis LP-153

Szintaktikus édesítőszerek — listák, egyebek

Listák alapstruktúra alakra hozása Farok-megadás betoldása.

[1,2] ⇒ [1,2|[]]. [[X|Y]] ⇒ [[X|Y]|[]]

Vessző (ismételt) kiküszöbölése [Elem1,Elem2...] ⇒ [Elem1|[Elem2...]].

[1,2|[]] ⇒ [1|[2|[]]]
[1,2,3|[]] ⇒ [1|[2,3|[]]] ⇒ [1|[2|[3|[]]]]

Struktúrakifejezéssé alakítás: [Fej|Farok] ⇒ .(Fej,Farok).

[1|[2|[]]] ⇒ .(1,.(2,[])), [[X|Y]|[]] ⇒ .(.(X,Y),[])

Egyéb szintaktikus édesítőszerek:

Karakterkód-jelölés: 0’Kar.

0’a ⇒ 97, 0’b ⇒ 98, 0’c ⇒ 99, 0’d ⇒ 100, 0’e ⇒ 101

Füzér (string): "xyz..." ⇒ az xyz... karakterek kódját tartalmazó lista

"abc" ⇒ [97,98,99], "" ⇒ [], "e" ⇒ [101]

Kapcsos zárójelezés: {Kif} ⇒ {}(Kif) (egy {} nevű, egyargumentumú struktúra — a {} jelpár egy önálló lexikai elem, egy névkonstans). Bináris, hexa stb. alak (csak iso módban), pl. 0b101010, 0x1a.

A Prolog szintaxis LP-154

Kifejezések szintaxisa — kétszintű nyelvtanok

Egy részlet egy „hagyományos” nyelv kifejezés-szintaxisából: ⟨ kifejezés ⟩ ::= ⟨ tag ⟩

| ⟨ kifejezés ⟩ ⟨ additív művelet ⟩ ⟨ tag ⟩ ⟨ tag ⟩ ::= ⟨ tényező ⟩

| ⟨ tag ⟩ ⟨ multiplikatív művelet ⟩ ⟨ tényező ⟩ ⟨ tényező ⟩ ::= ⟨ szám ⟩ | ⟨ azonosító ⟩ | ( ⟨ kifejezés ⟩ ) Ugyanez kétszintű nyelvtannal: ⟨ kifejezés ⟩ ::= ⟨ kif 2 ⟩ ⟨ kif N ⟩ ::= ⟨ kif N-1 ⟩

| ⟨ kif N ⟩ ⟨ N prioritású művelet ⟩ ⟨ kif N-1 ⟩ ⟨ kif 0 ⟩ ::= ⟨ szám ⟩ | ⟨ azonosító ⟩ | ( ⟨ kif 2 ⟩ ) {az additív ill. multiplikatív műveletek prioritása 2 ill. 1 }

A Prolog szintaxis LP-155

Kifejezések szintaxisa

⟨ programelem ⟩ ::= ⟨ kifejezés 1200 ⟩ ⟨ záró-pont ⟩

⟨ kifejezés N ⟩ ::= ⟨ op N fx ⟩ ⟨ köz ⟩ ⟨ kifejezés N-1 ⟩

| ⟨ op N fy ⟩ ⟨ köz ⟩ ⟨ kifejezés N ⟩
| ⟨ kifejezés N-1 ⟩ ⟨ op N xfx ⟩ ⟨ kifejezés N-1 ⟩
| ⟨ kifejezés N-1 ⟩ ⟨ op N xfy ⟩ ⟨ kifejezés N ⟩
| ⟨ kifejezés N ⟩ ⟨ op N yfx ⟩ ⟨ kifejezés N-1 ⟩
| ⟨ kifejezés N-1 ⟩ ⟨ op N xf ⟩
| ⟨ kifejezés N ⟩ ⟨ op N yf ⟩
| ⟨ kifejezés N-1 ⟩

⟨ kifejezés 1000 ⟩ ::= ⟨ kifejezés 999 ⟩ , ⟨ kifejezés 1000 ⟩

⟨ kifejezés 0 ⟩ ::= ⟨ név ⟩ ( ⟨ argumentumok ⟩ ) { A ⟨ név ⟩ és a ( közvetlenül egymás után áll!}

| ( ⟨ kifejezés 1200 ⟩ ) | { ⟨ kifejezés 1200 ⟩ }
| ⟨ lista ⟩ | ⟨ füzér ⟩
| ⟨ név ⟩ | ⟨ szám ⟩ | ⟨ változó ⟩

A Prolog szintaxis LP-156

Kifejezések szintaxisa — folytatás

⟨ op N T ⟩ ::= ⟨ név ⟩ {feltéve, hogy ⟨ név ⟩ N prioritású és T típusú operátornak lett deklarálva}

⟨ argumentumok ⟩ ::= ⟨ kifejezés 999 ⟩

| ⟨ kifejezés 999 ⟩ , ⟨ argumentumok ⟩ ⟨ lista ⟩ ::= []

| [ ⟨ listakif ⟩ ]

⟨ listakif ⟩ ::= ⟨ kifejezés 999 ⟩

| ⟨ kifejezés 999 ⟩ , ⟨ listakif ⟩
| ⟨ kifejezés 999 ⟩ | ⟨ kifejezés 999 ⟩

⟨ szám ⟩ ::= ⟨ előjeltelen szám ⟩

| + ⟨ előjeltelen szám ⟩
| - ⟨ előjeltelen szám ⟩

⟨ előjeltelen szám ⟩ ::= ⟨ természetes szám ⟩

| ⟨ lebegőpontos szám ⟩

A Prolog szintaxis LP-157

Kifejezések szintaxisa — megjegyzések

A ⟨ kifejezés N ⟩-ben ⟨ köz ⟩ csak akkor kell ha az őt követő kifejezés nyitó-zárójellel kezdődik.

| ?- op(500, fx, succ).
yes
| ?- write_canonical(succ (1,2)), nl, write_canonical(succ(1,2)).
succ(’,’(1,2))
succ(1,2)

A { ⟨ kifejezés ⟩ } azonos a {}(⟨ kifejezés ⟩) struktúrával, ez pl. a DCG nyelvtanoknál hasznos.

| ?- write_canonical({a}).
{}(a)

Egy ⟨ füzér ⟩ " jelek közé zárt karaktersorozat, általában a karakterek kódjainak listájával azonos.

| ?- write("baba").
[98,97,98,97]

A Prolog szintaxis LP-158

A Prolog lexikai elemei 1. (ismétlés)

⟨ név ⟩

kisbetűvel kezdődő alfanumerikus jelsorozat (ebben megengedve kis- és nagybetűt, számjegyeket és aláhúzásjelet);

egy vagy több ún. speciális jelből (+-*/\$^<>=‘~:.?@#&) álló jelsorozat; az önmagában álló ! vagy ; jel; a [] {} jelpárok;

idézőjelek (’) közé zárt tetszőleges jelsorozat, amelyben \ jellel kezdődő escape-szekvenciákat is elhelyezhetünk.

⟨ változó ⟩

nagybetűvel vagy aláhúzással kezdődő alfanumerikus jelsorozat.

az azonos jelsorozattal jelölt változók egy klózon belül azonosaknak, különböző klózokban különbözőeknek tekintődnek;

kivétel: a semmis változók (_) minden előfordulása különböző.

A Prolog szintaxis LP-159

A Prolog lexikai elemei 2.

⟨ természetes szám ⟩

(decimális) számjegysorozat;

2, 8 ill. 16 alapú számrendszerben felírt szám, ilyenkor a számjegyeket rendre a 0b, 0o, 0x karakterekkel kell prefixálni (csak iso módban) karakterkód-konstans 0’c alakban, ahol c egyetlen karakter ⟨ lebegőpontos szám ⟩ mindenképpen tartalmaz tizedespontot mindkét oldalán legalább egy (decimális) számjeggyel

e vagy E betűvel jelzett esetleges exponens

A Prolog szintaxis LP-160

Megjegyzések és formázó-karakterek

Megjegyzések (comment) A % százalékjeltől a sor végéig A /* jelpártól a legközelebbi */ jelpárig.

Formázó elemek

szóköz, újsor, tabulátor stb. (nem látható karakterek) megjegyzés A programszöveg formázása formázó elemek (szóköz, újsor stb.) szabadon elhelyezhetők; kivétel: struktúrakifejezés neve után nem szabad formázó elemet tenni; prefix operátor és ( közé kötelező formázó elemet tenni; ⟨ záró-pont ⟩: egy . karakter amit egy formázó elem követ.

TÍPUSOK PROLOGBAN

Típusok Prologban LP-162

Típusok leírása Prologban

Típusleírás: (tömör) Prolog kifejezések egy halmazának megadása Alaptípusok leírása: integer, float, number, atom, any Új típusok felépítése:

{ str(T_1, ..., T_n) } ≡ { str(e_1, ..., e_n) | e_1 ∈T_1, ..., e_n ∈T_n }, n ≥ 0

Példa: {személy(atom,atom,integer)} az olyan személy/3 funktorú struktúrák halmaza, amelyben az első két argumentum atom, a harmadik egész.

Típusok, mint halmazok úniója képezhető a \/ operátorral.

{személy(atom,atom,integer)} \/ {atom-atom} \/ atom

Egy típusleírás elnevezhető (kommentben): % :- type tnév == tleírás.

% :- type t1 == {atom-atom} \/ atom.,
% :- type ember == {ember-atom} \/ {semmi}.

Megkülönböztetett únió: csupa különböző funktorú összetett típus úniója. Egyszerűsített jelölés:

:- type T == { S_1 } \/ ...\/ { S_n }. ⇒ :- type T ---> S_1 ; ...; S_n.
  % :- type ember ---> ember-atom; semmi.
  % :- type egészlista ---> []; [integer|egészlista].

Típusok Prologban LP-163

Típusok leírása Prologban — folytatás

Paraméteres típusok — példák

% :- type list(T) ---> [] ; [T|list(T)]. % T típusú elemekből álló lista.
% :- type pair(T1, T2) ---> T1 - T2. % egy ’-’ nevű kétarg.-ú struktúra,
                                         % első arg. T1, a második T2 típusú.
% :- type assoc_list(KeyT, ValueT)
% == list(pair(KeyT, ValueT)). % KeyT és ValueT típusú
                                         % párokból álló lista.
% :- type szótár == assoc_list(szó, szó).
% :- type szó == atom.

Típusdeklarációk szintaxisa

⟨ típusdeklaráció ⟩ ::= ⟨ típuselnevezés ⟩ | ⟨ típuskonstrukció ⟩ ⟨ típuselnevezés ⟩ ::= :- type ⟨ típusazonosító ⟩ == ⟨ típusleírás ⟩ . ⟨ típuskonstrukció ⟩ ::= :- type ⟨ típusazonosító ⟩ ---> ⟨ megkülönb. únió ⟩ . ⟨ megkülönb. únió ⟩ ::= ⟨ konstruktor ⟩ ; . . .

⟨ konstruktor ⟩ ::= ⟨ névkonstans ⟩ | ⟨ struktúranév ⟩(⟨ típusleírás ⟩, . . . ) ⟨ típusleírás ⟩ ::= ⟨ típusnév ⟩ | ⟨ típusváltozó ⟩ | ⟨ típusleírás ⟩ \/ ⟨ típusleírás ⟩ | { ⟨ típusnév ⟩(⟨ típusleírás ⟩, . . . )} ⟨ típusazonosító ⟩ ::= ⟨ típusnév ⟩ | ⟨ típusnév ⟩(⟨ típusváltozó ⟩, . . . )

Típusok Prologban LP-164

Predikátum-deklarációk

Predikátumtípus-deklaráció

:- pred ⟨ eljárásnév ⟩(⟨ típusazonosító ⟩, ...)

Példák:

:- pred member(T, list(T)).
:- pred append(list(T), list(T), list(T)).

Predikátummód-deklaráció (Nem kötelező, több is megadható.)

:- mode ⟨ eljárásnév ⟩(⟨ módazonosító ⟩, ...) ahol ⟨ módazonosító ⟩ ::= in | out.

Példák:

:- mode append(in, in, in). % ellenőrzésre
:- mode append(in, in, out). % két lista összefűzésére
:- mode append(out, out, in). % egy lista szétszedésére

Vegyes típus- és móddeklaráció

:- pred ⟨ eljárásnév ⟩(⟨ típusazonosító ⟩::⟨ módazonosító ⟩, ...)

Példa:

:- pred between(integer::in, integer::in, integer::out).

Típusok Prologban LP-165

Móddeklaráció: a SICStus kézikönyv által használt alak

A SICStus kézikönyv egy másik jelölést használ a bemenő/kimenő argumentumok jelzésére, pl.

append(+L1, ?L2, -L3).
append(?L1, ?L2, +L3).

Mód-jelölő karakterek:

+ bemenő argumentum (behelyettesített)
- kimenő argumentum (behelyettesítetlen)
: eljárás-paraméter (meta-eljárásokban)
? tetszőleges